SAM-TR-72-10

COMPUTER PROGRAM FOR ACTIVITY DETERMINATIONS IN THE USAFSAM WHOLE-BODY COUNTER

TED D RUPP, CAPTAIN, USAF, BSC
ROBERT C. NELSON, CAPTAIN, USAF, BSC

USAF School of Aerospace Medicine
Aerospace Medical Division (AFSC)
Brooks Air Force Base, Texas

June 1972

Approved for public release; distribution unlimited.

43

Unclassified
Security Classification

DOCUMENT CONTROL DATA - R & D

(Security classification of title, body of abstract and indexing annotation must be entered when the overall report is classified)

1. ORIGINATING ACTIVITY (Corporate author)
USAF School of Aerospace Medicine
Aerospace Medical Division (AFSC)
Brooks Air Force Base, Texas 78235

2a. REPORT SECURITY CLASSIFICATION
Unclassified

2b. GROUP

3. REPORT TITLE

COMPUTER PROGRAM FOR ACTIVITY DETERMINATIONS IN THE USAFSAM WHOLE-BODY COUNTER

4. DESCRIPTIVE NOTES (Type of report and inclusive dates)
Final report January - May 1970

5. AUTHOR(S) (First name, middle initial, last name)
Ted D. Rupp, Captain, USAF, BSC
Robert C. Nelson, Captain, USAF, BSC

6. REPORT DATE	7a. TOTAL NO. OF PAGES	7b. NO. OF REFS
June 1972	38	4

8a. CONTRACT OR GRANT NO

b. PROJECT NO. 7757

c. Task No. 775701

d.

9a. ORIGINATOR'S REPORT NUMBER(S)
SAM-TR-72-10

9b. OTHER REPORT NO(S) (Any other numbers that may be assigned this report)

10 DISTRIBUTION STATEMENT

Approved for public release; distribution unlimited.

11 SUPPLEMENTARY NOTES

12. SPONSORING MILITARY ACTIVITY
USAF School of Aerospace Medicine
Aerospace Medical Division (AFSC)
Brooks Air Force Base, Texas 78235

13 ABSTRACT

The USAFSAM whole-body counter requires very precise calibration if it is to determine with accuracy the amounts of specific radioisotopes contained in a human body. Rigorous detection efficiencies for the isotopes of interest must be established in all channels of the instrument and frequently monitored. Calculations for accurate calibration and sample determination involve lengthy matrix operations which are too time-consuming for repetitive manual procedures. Presented is a computer program for the accurate computation of the amount of radioactive material, as measured by the USAFSAM whole-body counter, in a human body. The approach and development of the program and mathematical techniques are discussed, and all factors involved in the program are described in detail.

I

DD FORM 1 NOV 65 1473

Unclassified
Security Classification

14. KEY WORDS	LINK A		LINK B		LINK C	
	ROLE	WT	ROLE	WT	ROLE	WT
Whole-body counter Computer program Calibration Activity determinations						

IIa

COMPUTER PROGRAM FOR ACTIVITY DETERMINATIONS IN THE USAFSAM WHOLE-BODY COUNTER

TED D. RUPP, CAPTAIN, USAF, BSC
ROBERT C. NELSON, CAPTAIN, USAF, BSC

FOREWORD

This work was done in the Health Physics Branch, Radiobiology Division, under task No. 775701. The study was accomplished between January and May 1970. The paper was submitted for publication on 10 March 1972.

This report has been reviewed and is approved.

EVAN R. GOLTRA, Colonel, USAF, MC
Commander

ABSTRACT

The USAFSAM whole-body counter requires very precise calibration if it is to determine with accuracy the amounts of specific radioisotopes contained in a human body. Rigorous detection efficiencies for the isotopes of interest must be established in all channels of the instrument and frequently monitored. Calculations for accurate calibration and sample determination involve lengthy matrix operations which are too time-consuming for repetitive manual procedures. Presented is a computer program for the accurate computation of the amount of radioactive material, as measured by the USAFSAM whole-body counter, in a human body. The approach and development of the program and mathematical techniques are discussed, and all factors involved in the program are described in detail.

COMPUTER PROGRAM FOR ACTIVITY DETERMINATIONS

IN THE USAFSAM WHOLE-BODY COUNTER

I. INTRODUCTION

Constant vigilance is required to maintain an accurately calibrated whole-body counter. Mass suppression and efficiencies for the instrument must be continually remeasured and recalculated to compensate for component aging and electronic instabilities. The calibration procedure used is detailed by Dayton (1) and, in a somewhat briefer form, by Taboada (2). A generalization of the procedure will be presented to illustrate the data required by the computer program. The objective of the program is to simplify data manipulation after collection, thus making possible a daily or weekly update of calibration data points.

II. GAIN AND WINDOW SETTINGS

Calibrated radioactive sources were inserted in the counter. Gains and windows for each channel were adjusted to obtain maximum efficiency for the primary radionuclide and minimum efficiency for other radionuclides which might be present. Many criteria may be used to define which window settings are best, but final determination of the window settings can only be made by trial and error. (Further information on gain and window settings is given in references 1 and 2.)

III. MASS SUPPRESSION

The addition of mass in the whole-body counter causes an alteration in background. This alteration is a function of background energy, the Compton scattering cross section, and the mass of material in the counter. In the energy region above 0.5 Mev the background count rate tends to decrease as the amount of material is increased, while in the region from 0.1 to 0.5 Mev little or no change occurs. This effect is caused by high-energy gamma rays interacting with the material by Compton scattering, producing lower energy secondary radiation. The magnitude of the effect is a few percent of the net count rate from a normal man; therefore, it should be considered if accuracy to within a few percent is required. The effect has been included in the computer program.

Sugar phantoms free of gamma-emitting radionuclides were used to obtain mass suppression data, since sugar is easily obtained and handled and effectively approximates the elemental composition of a human being. A background count was taken before and after each phantom count to compensate for any short-range electronic instabilities. The number of times a phantom is counted for any given weight is left up to the discretion of the user. The main interest is to determine the regression

line between mass suppression and phantom weight for each channel. The method of least squares (3) is used to calculate the constants a and b in the equation $y = bx + a$.

Further explanation can be more easily accomplished mathematically. To do this some symbols need to be defined.

k = subscript denoting channel number being calibrated

j = subscript denoting phantom weight

i = subscript denoting an individual phantom count

w_{jk} = phantom weight, identified by channel

Y_{ijk} = gross count rate of phantom count, identified by channel

b_{1ijk} = background count rate before phantom count

b_{2ijk} = background count rate after phantom count

n_{jk} = number of counts taken of each phantom weight in a given channel

n_k = number of phantom weights for each channel

y_{jk} = average net count rate for phantom weight (w_{jk})

The average net count rate for w_{jk} is given by equation 1.

$$y_{jk} = \sum_{i=1}^{n_{jk}} \left(Y_{ijk} - \left(b_{1ijk} + b_{2ijk} \right)/2 \right) \Big/ n_{jk} \tag{1}$$

The slope b_k (c.p.m./lb.) and intercept a_k (c.p.m.) for each channel k are found by fitting the points (w_{jk}, y_{jk}). The formulas for a_k and b_k are given in equations 2 and 3.

$$b_k = \frac{n_k \left(\sum_{j=1}^{n_k} y_{jk} w_{jk} \right) - \left(\sum_{j=1}^{n_k} y_{jk} \right) \left(\sum_{j=1}^{n_k} w_{jk} \right)}{D_k} \tag{2}$$

and

$$a_k = \frac{(\sum_{j=1}^{n_k} w_{jk}^2)(\sum_{j=1}^{n_k} y_{jk}) - (\sum_{j=1}^{n_k} w_{jk})(\sum_{j=1}^{n_k} w_{jk} y_{jk})}{D_k} \tag{3}$$

where

$$D_k = n_k (\sum_{j=1}^{n_k} w_{jk}^2) - (\sum_{j=1}^{n_k} w_{jk})^2 \tag{4}$$

The standard deviations for b_k and a_k are given by equations 5 and 6.

$$S_{a_k} = (S_{y_{jk}|w_{jk}}) \left(\frac{\sum_{j=1}^{n_k} w_{jk}^2}{D_k} \right)^{1/2} \tag{5}$$

and

$$S_{b_k} = (S_{y_{jk}|w_{jk}}) \left(\frac{n_k}{D_k} \right)^{1/2} \tag{6}$$

where

$$S_{y_{jk}|w_{jk}} = \left(\frac{\sum_{j=1}^{n_k} (y_{jk} - (b_k w_{jk} + a_k))^2}{n_k - 2} \right)^{1/2} \tag{7}$$

which is the estimate of the standard error. The slopes and intercepts along with their standard deviations were stored in the computer memory for future use.

IV. EFFICIENCIES

The final step in the calibration is the data collection and computation of efficiencies for each channel. There will be n efficiencies for each channel—one efficiency for the radionuclide of primary interest, and an efficiency for each of the other radionuclides for which the counter is being calibrated. In all, there will be n^2 efficiencies when n channels are being calibrated.

The efficiencies and cross efficiencies are a function of the phantom weight, as was the background suppression. The gamma rays emitted from the calibration sources are absorbed and/or degraded by Compton scattering, the photoelectric effect and pair production resulting in their complete absorption or scattering into a lower energy window.

The data collection and data reduction for efficiency determination were very similar to those used for background suppression. Background counts were taken before and after each set of phantoms. The results were averaged and a best-fit linear regression line was computed by the method of least squares. The number of phantom counts was determined by the activity of the standards and statistics desired. The mathematical expressions were similar in form to the mass suppression equations with the addition of one more subscript. The symbols used are defined as follows:

k = subscript denoting channel number being calibrated

l = subscript denoting radionuclide

j = subscript denoting phantom weight

i = subscript denoting an individual phantom count

W_{jlk} = phantom weight, identified by channel

λ_l = decay constant for radionuclide

S_{jlk} = activity of standards corrected for decay

s_{jlk} = activity of standard at calibration date

E_{jlk} = efficiency

t_{jlk} = time from date of calibration to date of count

Y_{ijlk} = gross count rate of phantom

b_{1ijlk} = background count rate before phantom count

b_{2ijlk} = background count rate after phantom count

N_{jlk} = number of counts taken for each phantom weight in a given channel

N_{lk} = number of phantom weights for each channel

y_{jlk} = average net count rate for phantom weight W_{jlk}

The average net count rate for weight W_{jlk} is given by equation 8.

$$y_{jlk} = \sum_{i=1}^{N_{jlk}} \left(Y_{ijlk} - (b_{1ijlk} + b_{2ijlk})/2 + b_k W_{jlk} + a_k \right) \Big/ N_{jlk} \tag{8}$$

where $b_k W_{jlk} + a_k$ is the mass suppression correction term in the background. The equation for efficiency is

$$E_{jlk} = y_{jlk}/S_{jlk} \quad \text{where} \quad S_{jlk} = s_{jlk} \exp(-\lambda_l t_{jlk}) \tag{9}$$

The slope B_{lk} (eff./lb.) and intercept A_{lk} (eff.) are given by equations 10 and 11, respectively.

$$B_{lk} = \frac{N_{lk}\left(\sum_{j=1}^{N_{lk}} E_{jlk} W_{jlk}\right) - \left(\sum_{j=1}^{N_{lk}} E_{jlk}\right)\left(\sum_{j=1}^{N_{lk}} W_{jlk}\right)}{D_{lk}} \tag{10}$$

and

$$A_{lk} = \frac{\left(\sum_{j=1}^{N_{lk}} W_{jlk}^2\right)\left(\sum_{j=1}^{N_{lk}} E_{jlk}\right) - \left(\sum_{j=1}^{N_{lk}} W_{jlk}\right)\left(\sum_{j=1}^{N_{lk}} E_{jlk} W_{jlk}\right)}{D_{lk}} \tag{11}$$

where

$$D_{lk} = N_{lk}\left(\sum_{j=1}^{N_{lk}} W_{jlk}^{2}\right) - \left(\sum_{j=1}^{N_{lk}} W_{jlk}\right)^{2} \tag{12}$$

The standard deviations for B_{lk} and A_{lk} are given by equations 13 and 14.

$$S_{B_{lk}} = S_{E_{jlk}|W_{jlk}}\left(\frac{N_{lk}}{D_{lk}}\right)^{1/2} \tag{13}$$

$$S_{A_{lk}} = S_{E_{jlk}|W_{jlk}}\left(\frac{\sum_{j=1}^{N_{lk}} W_{jlk}^{2}}{D_{lk}}\right)^{1/2} \tag{14}$$

where

$$S_{E_{jlk}|W_{jlk}} = \left(\frac{\sum_{j=1}^{N_{lk}}\left(E_{jlk} - A_{lk}W_{jlk} - B_{lk}\right)^{2}}{N_{lk} - 2}\right)^{1/2} \tag{15}$$

is the estimate of the standard error.

V. COMPUTATION OF SUBJECT ACTIVITY

The basic equation for calculating the activity for two channels and two radionuclides is given by both Dayton and Taboada (1, 2). A straightforward generalization gives equation 16.

$$\begin{bmatrix} S_1 + MS_1 - (b_{11} + b_{12})/2 \\ S_2 + MS_2 - (b_{21} + b_{22})/2 \\ \vdots \\ S_k + MS_k - (b_{k1} + b_{k2})/2 \end{bmatrix} = \begin{bmatrix} E_{11} & E_{12} & \cdots & E_{1k} \\ E_{21} & E_{22} & \cdots & E_{2k} \\ \vdots & & & \vdots \\ E_{l1} & \cdot & \cdots & E_{l1} \end{bmatrix} \begin{bmatrix} A_1 \\ A_2 \\ \vdots \\ A_k \end{bmatrix} \tag{16}$$

where the symbols are defined as follows:

$MS_k = b_k W + a_k$

$E_{lk} = B_{lk} W + A_{lk}$

W = subject weight

S_k = subject gross count rate

b_{k1} = background count rate before subject count

b_{k2} = background count rate after subject count

A_k = activity present

The unknown in the equation is A_k. The solution can be found by using Cramer's rule (4). The resulting determinants are evaluated by the triangle method (4).

Even though the solution for A_k is straightforward, it requires the evaluation of a large number of determinants—an operation which is tedious and prone to error. The calculations are further complicated by the error analysis. Each time an operation in the evaluation of A_k is performed, the corresponding standard deviation is computed. This is done until the value and standard deviation for A_k are obtained. Three simple equations were used in approximating the standard deviations:

$$S = (s_1^2 + s_2^2)^{1/2} \text{—for addition and subtraction} \tag{17}$$

$$S = (x_1^2 s_2^2 + x_2^2 s_1^2)^{1/2} \text{—for multiplication} \tag{18}$$

and

$$S = \left(\left(\frac{s_1}{x_2}\right)^2 + \left(\frac{s_2 x_1}{x_2}\right)^2\right)^{1/2} \text{—for the division } (x_1/x_2). \tag{19}$$

VI. COMPUTER PROGRAM

The computer program consists of the mainline program which calculates mass suppression, efficiencies, and the table of normalized efficiencies versus weight. The mainline program also controls access to the subroutines. Four subroutines were established: (1) the subroutine BSSS, which calculates the activity and associated errors of the subject; (2) the subroutine DECAY, which decays the standards from time of calibration to present date; (3) the subroutine WLSCF, which performs the least-square curve fit on the mass suppression and efficiency data; and (4) the subroutine EDET, which takes determinants required in the other parts of the program.

A list of FORTRAN symbols, their mathematical equivalent, if any, and an explanation of their use are given in tables I through V. The same symbol may be used in several parts of the program. A complete FORTRAN listing of the program is in the appendix.

The data cards for the program are composed of eleven types. The information contained on each card type and the format for the types are given in table VI. The program deck is organized as shown in figure 1. Figure 2 illustrates the arrangement of a typical data deck for two channels.

The output of the program will vary depending upon the value used for ITAB. The two main items of interest, however, are the table of efficiencies and the subject activity.

The format of the efficiency table is controlled by card types 9 and 10. The variables are always printed in the same order. The weight is given first, then the mass suppression, the efficiencies, and the determinant of the efficiency matrix. The order of the printout corresponds to the sequence of input of the data to the computer.

The subject printout is fixed. An example is shown in figure 3.

The program was used to compute the activity of approximately one thousand subjects. The results of the program are very reliable and consistent with other methods of lean body mass and potassium determination. The errors for the subject activity are sensitive to any mistakes in the input data and appear to be a good measure of the accuracy of the subject activity.

REFERENCES

1. Dayton, R. R. The calibration and performance of the University of Florida whole-body counter. Master's thesis, University of Florida, Gainesville, Fla., Aug. 1965.

2. Taboada, J. Calibration and evaluation of the USAFSAM whole-body counter. SAM-TR-69-67, Nov. 1969.

3. Baird, D. C. Experimentation: An introduction to measurement theory and experiment design. Englewood Cliffs, N.J.: Prentice-Hall, 1965.

4. Scarborough, J. B. Numerical mathematical analysis, 4th ed. Baltimore, Md.: Johns Hopkins Press, 1958.

TABLE I

Symbol table for mainline program

FORTRAN symbol	Mathematical equivalent	Definition
ABACK(LS)	b_{1ijk} or $b_{1ij\|k}$	Background count rate before phantom, divided by 100
AIN(I,L)	$A_{\|k}$	Efficiency regression line intercept
AINT(I)	a_k	Mass suppression regression line intercept
BACK(LS)	b_{2ijk} or $b_{2ij\|k}$	Background count rate after phantom, divided by 100
D	y_{jk} or $y_{j\|k}$	Net count rate for phantom
DAY(IK)		Julian date of standard calibration
DSS(M)		Determinants of efficiencies
EDSS(M)		Standard deviation in DSS(M)
ERRI(I)	s_{a_k}	Standard deviation in AINT(I)
ERRORI(I,L)	$s_{A_{\|k}}$	Standard deviation in AIN(I,L)
ERRORS(I,L)	$s_{B_{\|k}}$	Standard deviation in SLOPE(I,L)
ERRS(I)	s_{b_k}	Standard deviation in SLOP(I)
F		Equal to NUMB(J) or NUMP(N)
FORM		Format for efficiency table
FORS		Format for errors in efficiency table
HD(KN)		Heading for efficiency table
ID		Equal to NUMB(J)
IDUM		Counter
IDUV		Counter
IDUS		Counter

TABLE I (contd.)

FORTRAN Symbol	Mathematical equivalent	Definition
IDUT		Counter
IF		Counter
IG		Counter
IK		Counter
IR		Equal to NUMP(N)
IS		Equal to NUMBP(N)
ITAB		Print and computation control
IU		Counter
IX		Equal to NUMPO(I)
J		Counter
K		Counter
KJ		Counter
KSSD		Counter
KZZ		Counter
L		Counter
LAMBDA	λ_l	Decay constant for standards
LS		Counter
N		Counter
NN		Counter
NUMB(J)	n_{jk}	Number of counts taken for a given phantom weight in channel k
NUMBC	n	Number of channels being calibrated

TABLE I (contd.)

FORTRAN symbol	Mathematical equivalent	Definition
NUMBP(N)	n_k	Number of phantom weights per channel
NUMP(N)	N_{jlk}	Number of data points for phantom weight WGTS(J)
NUMPO(IK)	N_{lk}	Number of phantom weights used to determine efficiency
PDAY(IK)		Present Julian date
POIN(LS)	Y_{ijk} or Y_{ijlk}	Gross count rate of phantom divided by 100
PYEAR(IK)		Last two digits of present year
SKR(IF,IG)		Efficiency matrix for weight SW
SKT(IF,IG)		Error matrix for weight SW
SLOP(N)	b_k	Slope of mass suppression regression lines
SLOPE(N,L)	B_{lk}	Slope of efficiency regression line
STD(IK)	s_{jlk}	Standard at time of calibration
STDD(J)	S_{jlk}	Standard corrected for decay
SW		Weight in pounds for TABLE(I,J)
TABLE(I,J)		Table of efficiencies and mass suppression as a function of weight
WGT(J)	w_{jk}	Phantom weights used in mass suppression
WGTS(J)	W_{jlk}	Phantom weights used in efficiencies
YEAR(IK)		Last two digits of year source was calibrated
YS(J)	y_{jk}	Average net count rate for phantom weight WGT(J)
YSS(J)	y_{jlk}	Average net count rate for phantom weight WGTS(J)

TABLE II

Symbol table for subroutine BSSS

FORTRAN symbol	Mathematical equivalent	Definition
ABACK(I)	b_{k2}	Background count rate after subject, divided by 100
BACK(I)	b_{k1}	Background count rate before subject, divided by 100
CW(K)		Standard deviation error in XS(K)
DSS(M)		Determinant of efficiencies
DW(K)		Standard deviation in S(K)
EB(IL)		Standard deviation in average background
EDSS(M)		Standard deviation in DSS(M)
ES(IL)		Standard deviation in SOURCE(I)
IJ		Counter
IL		Counter
IM		Counter
ISUB		IWGT-70
IWGT		Subject's weight rounded to the nearest pound.
J		Counter
K		Counter
M		Counter
N		Counter
NCHAN(I)	k	Channel number
NODE		Designates if subject's weight is in pounds or kilograms
NUDE		Last card indicator

TABLE II (contd.)

FORTRAN symbol	Mathematical equivalent	Definition
NUMBC	n	Number of channels being calculated
NUMS		Subject sample number
S(K)	A_k	Disintegrations per minute, divided by 100, for channel k
SOURC(I)	S_k	Gross count rate of subject, divided by 100
SOURCE(I)		Net count rate of subject
TABLE(ISUB, IJ)		Table of mass suppression efficiencies and their standard deviations
TIME		Length of time for count
WGT	W	Subject's weight in pounds or kilograms
X(N,M)	E_{lk}	Efficiency matrix for subject
XS(K)		Determinant of Z(K,N)
XX(N,M)		Working storage
Y(N,M)		Standard deviation matrix for X(N,M)
Z(K,N)		Working storage

TABLE III

Symbol table for subroutine DECAY

FORTRAN symbol	Mathematical equivalent	Definition
CCURIE	S_{jlk}	Source strength corrected for decay
CURIE	s_{jlk}	Source strength at time of calibration
DAY		Julian date source was calibrated
LAMBDA	λ_l	Decay constant
PDAY		Present Julian date
PYEAR		Last two digits of present year
XPDAY		Number of days between PDAY and DAY
XPYEAR	t_{jlk}	Elapsed time between calibration date and present date
YEAR		Last two digits of year source was calibrated

TABLE IV

Symbol table for subroutine WLSCF

FORTRAN symbol	Mathematical equivalent	Definition
AINTER	b	Y intercept of the regression line
DELTA		Working storage
DIFF		Difference between experimental and calculated value of Y(I)
ERRORI	S_b	Standard deviation of AINTER
ERRORS	S_a	Standard deviations in SLOPE
I		Counter
N		Counter
SLOPE	a	Slope of the regression line
SN		Working storage
SUMX	$\sum_{i=1}^{10} X_i$	Sum of X(I)'s
SUMXY	$\sum_{i=1}^{10} X_i Y_i$	Sum of the product X(I) and Y(I)
SUMY	$\sum_{i=1}^{10} Y_i$	Sum of Y(I)'s
SY		
X(I)	X_i	Independent variable
Y(I)	Y_i	Dependent variable

TABLE V

Symbol table for subroutine EDET

FORTRAN symbol	Mathematical equivalent	Definition
B(KS,K)		Reduced matrix
C	a_{ij}/a_{jj}	Reducing multiplier
D	$\mathrm{Det}\lvert a_{ij}\rvert$	Determinant of S(KS,K)
ERB(KS,K)		Standard deviation in B(KS,K)
ERB2		Variance in B(KS,K)
ERD	S_D	Standard deviation in D
ERD2		Working storage
ERC2		Standard deviation in C
ES(KS,K)	$S_{a_{ij}}$	Standard deviation matrix for S(KS,K)
J		Order of matrix S(KS,K)
JK		Counter
JS		Counter
K		Counter
KI		Counter
KR		Counter
KS		Counter
KT		Counter
L		Counter
S(KS,K)	a_{ij}	Matrix to be evaluated

TABLE VI

Input data cards

Type	Columns	Variables	Format	Remarks
1	1-50		50H	Title
	51-80			Unused
2	1-3	NUMBC	I3	NUMBC$\leq$5
	4-6	NUMP(1)	I3	
	7-9	NUMP(2)	I3	
	10-12	NUMP(3)	I3	NUMBP(I)$\leq$10
	13-15	NUMP(4)	I3	
	16-18	NUMP(5)	I3	
	19-21	ITAB(*)	I3	Unused
	22-80			Unused
3	1-4	WGT(1)	F4.0	
	5-7	NUMB(1)	I3	NUMB(I)$\leq$100
	8-11	WGT(2)	F4.0	
	12-14	NUMB(2)	I3	
	15-18	WGT(3)	F4.0	
				Repeats until subscript reaches NUMBP(I)
4	1-9		9X	Unused
	10-16	POIN(LS)	F7.1	
	17-23	BACK(LS)	F7.1	
	24-30	ABACK(LS)	F7.1	
	31-70		40H	Remarks unused
	71-80			

TABLE VI (contd.)

Type	Columns	Variables	Format	Remarks
5	1-3	NUMPO(1)	I3	
	4-6	NUMPO(2)	I3	
	7-9	NUMPO(3)	I3	NUMPO(I) $\leq$10
	10-12	NUMPO(4)	I3	
	13-15	NUMPO(5)	I3	
	16-22	LAMBDA	F7.6	Floating point
	23-80			Unused
6	1-4	WGTS(1)	F4.0	
	5-11	STD(1)	F7.2	
	12-13	YEAR(1)	F2.0	
	14-17	DAY(1)	F4.0	
	18-20	NUMP(1)	I3	NUMP(I) $\leq$100
	21-24	WGTS(2)	F4.0	
	25-31	STD(2)	F7.2	
	32-33	YEAR(2)	F2.0	
	34-37	DAY(2)	F4.0	Repeats until subscript reaches NUMPO(I)
	38-40	NUMP(3)	I3	
7	1-9		9X	Unused
	10-16	POIN(LS)	F7.1	
	17-23	BACK(LS)	F7.1	
	24-30	ABACK(LS)	F7.1	
	31-48		18X	Unused
	49-51	PYEAR(LS)	F3.0	
	52-54		3X	Unused

TABLE VI (contd.)

Type	Columns	Variables	Format	Remarks
7	55-57	PDAY(LS)	F3.0	
	58-80			Unused
8	1-80	†	13A6,A2	Title card for efficiency table
9	1-60	FORM or FORS†	10A6	Format for printout of efficiency table
10	1-78	‡	78H	Subject name, SSAN, date, etc.
11	1-4	NUMS ‡	I4	
	5-12	WGT	F8.5	
	13	NODE	I1	
	14-22	TIME	F9 4	
	23	NCHAN(1)	I1	NCHAN(I)≥NUMBC
	24-30	SOURCE(1)	F7.1	
	31-37	BACK(1)	F7.1	
	38-44	ABACK(1)	F7.1	
	45	NCHAN(2)	I1	Repeats until subscript reaches NUMBC

*ITAB controls the function of the program according to the following conditions:

MOD(ITAB,2)≥0	Print table of efficiencies as a function of weight
MOD(ITAB,4)≥2	Calculate activity of subjects
MOD(ITAB,16)≥8	Print slopes and intercepts for mass suppression and standards
MOD(ITAB,32)≥16	Print mass suppression data
MOD(ITAB,64)≥32	Print standard calibration data

†Input cards not required unless MOD(ITAB,2)≥0

‡Input cards not required unless MOD(ITAB,4)≥2



FIGURE 1

Organization of program deck.



FIGURE 2

Example of data deck for calibration of two channels.

DOE, JOHN E. JR.[a]	000000000[b]			1 JAN 70[c]
0000[d]	200.0[e]	CHANNEL NO.	DPM/100	ERROR/100
		1	186.85[f]	18.92[g]
		2	268.02	27.24

[a] Subject's name.

[b] Social security number.

[c] Date count was made.

[d] Subject's code number.

[e] Weight in pounds.

[f] Activity divided by 100. The units on the activity are the same as those on the standards. They must be consistent throughout the program.

[g] Standard deviation in the activity, divided by 100.

FIGURE 3

Example of subject output.

APPENDIX

FORTRAN LISTING OF

WHOLE-BODY COUNTER PROGRAM

Preceding page blank

```
      DIMENSION  NUMB(10),WGT(10),POIN(100),BACK(100),ABACK(100),
     1NUMBP(5),YS(10),DAY(10),YSS(10),YEAR(10),SLOP(5),ERRS(5),AINT(5),
     2ERRI(5),STD(5),NUMPO(5),NUMP(10),WGTS(10),SLOPE(5,5),AIN(5,5),
     3FORM(10),FORS(10),PYEAR(100),PDAY(100),SKR(5,5),SKT(5,5),HD(21),
     4STDD(10),EDSS(221),DSS(221),TABLE(221,61),ERRORS(5,5),ERRORI(5,5)
      REAL LAMBDA
      READ 1000
      PRINT 1000
C. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
C      INITIALIZATION
C. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
      DO 100 NN =1,221
      DO 100 N  =1,61
  100 TABLE(NN,N)=0.0
      DO 103 N=1,5
      NUMBP(N)=0.0
      SLOP(N) =0.0
      ERRS(N) =0.0
      AINT(N) =0.0
      ERRI(N) =0.0
      STD(N)  =0.0
      NUMPO(N)=0.0
      DO 103 NN =1,5
      ERRORS(N,NN)=0.0
      SLOPE(N,NN) =0.0
      AIN(N,NN)=0.0
  103 ERRORI(N,NN)=0.0
      DO 102 NN =1,100
      EDSS(NN) =0.0
      PYEAR(NN) =0.0
      PDAY(NN) =0.0
      POIN(NN) =0.0
      BACK(NN) =0.0
  102 ABACK(NN)=0.0
      DO 105 N =1,10
      YSS(N) =0.0
```

Preceding page blank

```
      STDD(N) =0.0
      DAY(N) =0.0
      YEAR(N) =0.0
      WGT(N) =0.0
      NUMB(N)=0.0
      YS(N)  =0.0
      NUMP(N)=0.0
  105 WGTS(N)=0.0
C. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
C     MASS SUPPRESSION CALCULATIONS
C. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
      READ 1001,NUMBC,(NUMBP(I),I=1,NUMBC),ITAB
      IF(MOD(ITAB,32).LT.16) GO TO 60
      PRINT 1001,NUMBC,(NUMBP(I),I=1,NUMBC)
   60 DO 12 I=1 ,NUMBC
      IS=NUMBP(I)
      READ 1002,(WGT(J),NUMB(J),J=1,IS)
      IF(MOD(ITAB,32).LT.16) GO TO 61
      PRINT 1002,(WGT(J),NUMB(J),J=1,IS)
   61 DO 10 J=1,IS
      ID=NUMB(J)
      READ 1003,(POIN(LS),BACK(LS),ABACK(LS),LS=1,ID)
      IF(MOD(ITAB,32).LT.16) GO TO 62
      PRINT 1003,(POIN(LS),BACK(LS),ABACK(LS),LS=1,ID)
   62 YS(J) =0.0
      F =NUMB(J)
      DO 11 K=1,ID
      D=-(POIN(K)-(BACK(K)+ABACK(K))/2.0)*100.0
   11 YS(J) =YS(J) +D
      YS(J) =YS(J)/F
   10 CONTINUE
      CALL WLSCF(WGT,YS,SLOP(I),ERRS(I),AINT(I),ERRI(I))
      IF(MOD(ITAB,16).LT. 8) GO TO 12
      PRINT 1006,SLOP(I),ERRS(I),AINT(I),ERRI(I)
   12 CONTINUE
      DO 20 I=1,221
```

```
      SW = I + 70
      TABLE(I,1) =SW
      DO 20 N=1,NUMBC
      TABLE(I,2*N) =SLOP(N)*SW +AINT(N)
   20 TABLE(I,2*N+1)=SQRT((ERRS(N)*SW)**2+ERRI(N)**2)
C. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
C     CALCULATION OF EFFICIENCY
C. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
      L=1
   21 READ 1004,(NUMPO(IK),IK=1,5     ),LAMBDA
      IF(MOD(ITAB,64).LT.32) GO TO 68
      PRINT 1004,(NUMPO(IK),IK=1,5),LAMBDA
   68 DO 30 I=1,NUMBC
      IF(LAMBDA.LT.0.000010) LAMBDA =0.0
      IX = NUMPO(I)
      DO 73 II =1,10
   73 STDD(II) =0.0
      READ 1005,(WGTS(J),STD(J),YEAR(J),DAY(J),NUMP(J)  ,J=1,IX)
      IF(MOD(ITAB,64).LT.32) GO TO 69
      PRINT 1005,(WGTS(J),STD(J),YEAR(J),DAY(J),NUMP(J),J=1,IX)
   69 DO 15 J=1,IX
      STD(J)=STD(J)*100.
      IR =NUMP(J)
      READ  1008,(POIN(LS),BACK(LS),ABACK(LS),PYEAR(LS),PDAY(LS),LS=1,
     1IR)
      IF(MOD(ITAB,64).LT.32) GO TO 64
      PRINT 1008,(POIN(LS),BACK(LS),ABACK(LS),PYEAR(LS),PDAY(LS),LS=1,
     1IR)
   64 YSS(J) =0.0
      F =NUMP(J)
      DO 16 K =1,IR
      POIN(K)=POIN(K)*100.
      BACK(K)=BACK(K)*100.
      ABACK(K)=ABACK(K)*100.
      CALL DECAY(STD(J),YEAR(J),DAY(J),PYEAR(K),PDAY(K),LAMBDA,STDD(J))
      D=(POIN(K)-(BACK(K)+ABACK(K))/2.0+SLOP(I)*WGTS(J)+AINT(I))/STDD(J)
```

```
   16 YSS(J) =YSS(J) +D
      YSS(J) =YSS(J)/F
   15 CONTINUE
      CALL WLSCF(WGTS,YSS,SLOPE(I,L),ERRORS(I,L),AIN(I,L),ERRORI(I,L))
      IF(MOD(ITAB,16).LT. 8) GO TO 30
      PRINT 1006,SLOPE(I,L),ERRORS(I,L),AIN(I,L),ERRORI(I,L)
   30 CONTINUE
      L =L+1
      IF(L.LE.NUMBC) GO TO 21
      IS=2*NUMBC+2
      IU =IS +NUMBC**2
      ID =NUMBC**2 -1
      DO 31 M =1,221
      SW = M+70
      DO 65 IG =1,NUMBC
      DO 65 IF =1,NUMBC
      SKR(IF,IG) =SLOPE(IF,IG)*SW +AIN(IF,IG)
   65 SKT(IF,IG)= SQRT((ERRORS(IF,IG)*SW)**2 + ERRORI(IF,IG)**2)
      CALL EDET(NUMBC, SKR,SKT, DSS(M),EDSS(M))
      KSSD =IS-2
      DO 31 I=1,NUMBC
      DO 31 J=1,NUMBC
      KSSD = KSSD +2
      TABLE(M,KSSD)= SLOPE(J,I)*SW+AIN(J,I)
   31 TABLE(M,KSSD+1) = SQRT((ERRORS(J,I)*SW)**2 + ERRORI(J,I)**2)
      KZZ= 1
      IF(MOD(ITAB, 2).EQ. 0) GO TO 66
C. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
C     TABLE PRINTOUT
C. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
   50 IF(NUMBC.GT.3) GO TO 48
      IDUM =2
      IDUN=(NUMBC+NUMBC**2)*2 +1
      IDUS =IDUN +1
      IDUT =IDUM +1
      GO TO 49
```

```
48 IF(KZZ.GT.1) GO TO 51
   IDUN = 30
   IDUS = 31
   IDUT = 3
   IDUM = 2
   GO TO 49
51 IDUN =2*(NUMBC + NUMBC**2) +IDUN
   IDUS =IDUN +1
   IDUM = 32
   IDUT = 33
49 READ  1007,(HD(KN),KN=1,21)
   READ 1010,FORM
   READ 1010,FORS
   N =10
   L = 2
   M = 1
   PRINT 1015,(HD(KN),KN=1,21)
44 DO 45 KJ= M,N
   IF(L.LT.60) GO TO 46
   PRINT 1015,(HD(KN),KN=1,21)
   L =2
46 PRINT FORM,TABLE(KJ,1),(TABLE(KJ,KC),KC =IDUM,IDUN,2),DSS(KJ)
45 L =L+1
   IF(L.LT.60) GO TO 80
   PRINT 1015,(HD(KN),KN=1,21)
   L =2
80 PRINT FORS,(TABLE(N,KC),KC = IDUT,IDUS,2),EDSS(N)
   M = M +10
   N = M +9
   L = L +1
   IF(N.LT.221) GO TO 44
   IF(NUMBC.LE.3) GO TO 47
   KZZ =2
   GO TO 50
47 CONTINUE
66 IF(MOD(ITAB,4).LT.2) GO TO 67
```

```
      CALL BSSS(DSS,EDSS,TABLE,NUMBC)
   67 CONTINUE
 1000 FORMAT(50H                                                  )
 1001 FORMAT(7I3)
 1002 FORMAT(10(F4.0,I3))
 1003 FORMAT(9X,3F7.1,40H                                        )
 1004 FORMAT(5I3,F7.6)
 1005 FORMAT(F4.0,F7.2,F2.0,F4.0,I3,F4.0,F7.2,F2.0,F4.0,I3,F4.0,F7.2,
     1F2.0,F4.0,I3,F4.0,F7.2,F2.0,F4.0,I3)
 1006 FORMAT(1H ,4F10.6)
 1007 FORMAT(13A6,A2)
 1008 FORMAT(9X,3F7.1,18X,F3.0,3X,F3.0)
 1010 FORMAT(10A6)
 1015 FORMAT(1H1,13A6,A1,6A6,A5)
      STOP
      END
```

```
      SUBROUTINE BSSS(DSS,EDSS,TABLE,NUMBC)
C. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
C     CALCULATION OF SUBJECTS ACTIVITY
C. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
      DIMENSION X(5,5),Y(5,5),Z(5,5),EDSS(221),NCHAN(5),XX(5,5),BACK(5),
     1ABACK(5),SOURCE(5),SOURC(5) ,EB(5),ES(5),XS(5),S(5),DW(5),CW(5),
     2DSS(221),TABLE(221,61)
      PRINT 1005
 1005 FORMAT(1H1)
   80 DO 10 I =1,5
      NCHAN(I)=0
      BACK(I) =0.0
      ABACK(I)=0.0
      SOURC(I)=0.0
      SOURCE(I)=0.0
      ES(I)=0.0
      EB(I)=0.0
      XS(I)=0.0
      S(I) =0.0
      DW(I)=0.0
      CW(I)=0.0
      DO 10 J=1,5
      X(J,I)=0.0
      Y(J,I)=0.0
   10 Z(J,I)=0.0
      AS = 0.0
      READ 1001
      PRINT 1001
      READ 1000,NUMS,WGT,NODE,TIME,(NCHAN(I),SOURC(I),BACK(I),ABACK(I),
     1I=1,2),NUDE
      IF(NUMBC.LE.2) GO TO 81
      READ 1004,(NCHAN(I),SOURC(I),BACK(I),ABACK(I),I=3,NUMBC)
   81 IF(NODE.EQ.0) GO TO 11
      WGT =WGT*2.205
   11 IWGT = WGT+0.5
      ISUB =IWGT -70
```

```
      DO 12 I =1,NUMBC
      IF(NCHAN(I).EQ.0) GO TO 12
      IJ = 2*NCHAN(I)
      IL = NCHAN(I)
      SOURC(I)=SOURC(I)*100.
      BACK(I)=BACK(I)*100.
      ABACK(I)=ABACK(I)*100.
      EB(IL)=(BACK(I)+ABACK(I))/(4.0*TIME)
      SOURCE(IL)=SOURC(I)-(BACK(I)+ABACK(I))/2.0 +TABLE(ISUB,IJ)
      ES(IL)=SOURC(I)/TIME
      ES(IL)=SQRT(ES(IL)+EB(IL)   + TABLE(ISUB,IJ+1)**2)
   12 CONTINUE
      IM=2*NUMBC
      DO 13 N =1,NUMBC
      DO 13 M =1,NUMBC
      IM = IM +2
      X(N,M) =TABLE(ISUB,IM)
   13 Y(N,M)=TABLE(ISUB,IM+1)
      DO 51 K=1,NUMBC
      DO 50 J=1,NUMBC
      DO 50 I=1,NUMBC
      XX(I,J) = Y(I,J)
   50 Z(I,J)= X(I,J)
      DO 52 N=1,NUMBC
      XX(K,NN) = ES(N)
   52 Z(K,N)=SOURCE(N)
      CALL EDET(NUMBC,Z, XX,XS(K),CW(K))
   51 S(K)=XS(K)/DSS(ISUB)
      DO 63 K=1,NUMBC
      DW(K) =SQRT((CW(K)/DSS(ISUB))**2 +(XS(K)*EDSS(ISUB)/(DSS(ISUB)**2)
     1)**2)
      DW(K)=DW(K)/100.
   63 S(K)=S(K)/100.
      PRINT 1002,NUMS,WGT
      DO 70 I = 1,NUMBC
   70 PRINT 1003,I,S(I),DW(I)
```

```
      IF(NUDE.EQ.0)GO TO 80
1000 FORMAT(I4,F8.5,I1,F9.4,I1,3F7.1,I1,3F7.1,I1)
1001 FORMAT(78H
    1                             )
1002 FORMAT(25X,I5,1X,F5.1,35H   CHANNEL NO.  DPM*100    ERROR*100 )
1003 FORMAT(42X,I1,6X,F10.2,2X,F10.4)
1004 FORMAT(3(I1,3F7.1))
      RETURN
      END
```

```
      SUBROUTINE DECAY(CURIE,YEAR,DAY,PYEAR,PDAY,LAMBDA,CCURIE)
C. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
C     DECAYS STANDARDS
C. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
      REAL LAMBDA
      IF(PDAY -DAY)30,40,50
   30 XPDAY =PDAY +365. -DAY
      PYEAR =PYEAR -1.0
      GO TO 60
   40 XPDAY  =0.0
      GO TO 60
   50 XPDAY =PDAY -DAY
   60 CONTINUE
      XPYEAR =0.0
      XPYEAR =(PYEAR -YEAR)* 365. +XPDAY
      CCURIE = CURIE *EXP((-LAMBDA)*XPYEAR)
      RETURN
      END
```

```
      SUBROUTINE WLSCF(X,Y,SLOPE,ERRORS,AINTER,ERRORI)
C. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
C     LEAST SQUARE CURVE FIT
C. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
      DIMENSION X(10),Y(10)
      SUMX=0.0
      SUMX2=0.0
      SUMY=0.0
      SUMXY=0.0
      N =0
      DIFF=0.0
      DO 1000 I=1,10
      IF(X(I).EQ.0.0.AND.Y(I).EQ.0.0)GO TO 1000
      N=N+1
      SUMX=SUMX+X(I)
      SUMY=SUMY+Y(I)
      SUMX2=SUMX2+X(I)**2
      SUMXY=SUMXY+X(I)*Y(I)
 1000 CONTINUE
      SN=N
      DELTA = SN*SUMX2-SUMX**2
      SLOPE =(SN*SUMXY-SUMY*SUMX)/DELTA
      AINTER=(SUMX2*SUMY -SUMX*SUMXY)/DELTA
      DO 1001 I=1,10
      IF(X(I).EQ.0.0.AND.Y(I).EQ.0.0)GO TO 1001
      DIFF = DIFF+(Y(I)-SLOPE*X(I)-AINTER)**2
 1001 CONTINUE
      SY=SQRT(DIFF   /(SN-2.0))
      ERRORS=SY*SQRT(SN/DELTA)
      ERRORI=SY*SQRT(SUMX2/DELTA)
      RETURN
      END
```

```
      SUBROUTINE EDET(J,S,ES,D,ERD)
C. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
C     CALCULATES ERROR IN DETERMINANTS
C. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
      DIMENSION S(5,5),ES(5,5),B(5,5),ERB(5,5)
      JK=J-1
      DO 11 KI =1,JK
      DO 10 K =KI,J
      KT = KI+1
      IF(  S(KI,KI).EQ.0.0) GO TO 15
      DO 10 KS =KT,J
      C =S(KS,KI)/S(KI,KI)
      ERC2=  (ES(KS,KI)/S(KI,KI))**2 + (S(KS,KI)*ES(KI,KI)/S(KI,KI)**2)*
     1*2
      ERD2 = (S(KI,K)**2)*ERC2 +(C*ES(KI,K))**2
      ERB2 =  ES(KS,K)**2 + ERD2
      ERB(KS,K) =SQRT(ERB2)
   10 B(KS,K) =S(KS,K)-C*S(KI,K)
   15 DO 11 KR= KT,J
      DO 11 JS= KI,J
      ES(KR,JS) =ERB(KR,JS)
   11 S(KR,JS) =B(KR,JS)
      D=1.0
      ERD =0.0
      DO 12 L = 1,J
      ERD  = SQRT((D* ES(L,L))**2 +(ERD* S(L,L))**2)
   12 D= S(L,L)*D
      RETURN
      END
```